A MATHEMATICA PACKAGE FOR COMPUTING ASYMPTOTIC EXPANSIONS OF SOLUTIONS OF P-FINITE RECURRENCE EQUATIONS

MANUEL KAUERS *

Abstract. We describe a simple package for computing a fundamental system of certain formal series solutions, up to a prescribed order, of a given P-finite recurrence equation. These solutions can be viewed as describing the asymptotic behavior of sequences satisfying the recurrence.

1. The Problem

The Mathematica code described below solves the following problem:

- **Given** a linear recurrence equation of order r with polynomial coefficients (also known as a P-finite recurrence)

$$p_0(n)a_n + p_1(n)a_{n+1} + \cdots + p_r(n)a_{n+r} = 0,$$

- **Find** r linearly independent solutions of the form

$$\begin{aligned} n^{\gamma n} e^{u(n^{1/r})} \rho^n n^\alpha \Big(&\big(0 + \frac{\beta_{0,1}}{n} + \frac{\beta_{0,2}}{n^2} + \frac{\beta_{0,3}}{n^3} + \cdots\big) \\ &+ \big(0 + \frac{\beta_{1,1}}{n} + \frac{\beta_{1,2}}{n^2} + \frac{\beta_{1,3}}{n^3} + \cdots\big)\log(n) \\ &+ \cdots \\ &+ \big(1 + \frac{\beta_{k-1,1}}{n} + \frac{\beta_{k-1,2}}{n^2} + \frac{\beta_{k-1,3}}{n^3} + \cdots\big)\log(n)^{k-1}\Big) \end{aligned}$$

  where $\alpha, \gamma, \rho, \beta_{i,j}$ are constants, r, s, k are positive integers, and u is a polynomial.

It is well known and not difficult to compute this data [6], and besides our implementation described below, there are several others which do the same job [7, 1]. The Mathematica code I programmed a couple of years ago was originally intended for private use only, but over the time, I have also given copies to other users, and this often raised the question on how to use the package and how exactly to interpret its output. The purpose of this technical report is to answer these questions.

A priori, the output is only correct in some formal algebraic sense. But according to Birkhoff and Trjitzinsky [2], it is also correct analytically in the sense that every sequence (a_n) satisfying the input recurrence has a linear combination of the output series as its asymptotic

* Supported by the Austrian FWF grant Y464-N18.

expansion. This means in particular that we can obtain asymptotic estimates of the form

$$a_n = cn^{\gamma n} \mathrm{e}^{u(n^{1/r})} \rho^n n^\alpha \Big(\big(0 + \frac{\beta_{0,1}}{n} + \cdots + \frac{\beta_{0,N}}{n^N} + \mathrm{O}\big(\frac{1}{n^{N+1}}\big)\big) + \cdots + \big(1 + \frac{\beta_{k-1,1}}{n} + \cdots + \frac{\beta_{k-1,N}}{n^N} + \mathrm{O}\big(\frac{1}{n^{N+1}}\big)\big) \log(n)^{k-1} \Big) \quad (n \to \infty)$$

for any prescribed N and some constant c which can be determined numerically to high accuracy. Then also

$$a_n \sim cn^{\gamma n} \mathrm{e}^{u(n^{1/r})} \rho^n n^\alpha \log(n)^{k-1} \qquad (n \to \infty)$$

where $a_n \sim b_n$ means asymptotic equivalence in the sense that $\lim_{n\to\infty} \frac{a_n}{b_n} = 1$.

Unlike the "algebraic correctness", the "analytic correctness" is not easy to show. It should be remarked that the arguments given by Birkhoff and Trjitzinsky are long and complicated, and that some people hesitate to believe in them. On concrete examples, however, I have never observed any mismatch between the actual asymptotic behavior of a solution (a_n) and the formal expansions.

An alternative approach, whose underlying analytic theory is more widely accepted but which would require more programming effort, is to go via the generating function $A(z) := \sum_{n=0}^{\infty} a_n z^n$ of the sequence (a_n) under consideration. Flajolet and Sedgewick [3] give a comprehensive account on the correspondence between the asymptotic behavior of $A(z)$ near its singularities closest to the origin and the asymptotic behavior of the sequence (a_n). Salvy's Maple package gdev [5] follows this approach.

2. The Package

The package is available for download at the URL

http://www.risc.jku.at/research/combinat/software/

It was written and tested for Mathematica 6 and 7.

Two commands are provided by the package. The first, **Asymptotics,** takes a recurrence as input and returns the dominant term of all its (formal) asymptotic solutions.

In[1]:= **<< Asymptotics.m**

Asymptotics Package by Manuel Kauers – © RISC Linz – V 0.3 (2011-03-31)

In[2]:= **Asymptotics[$(n+1)f[n+2] - (3n+2)f[n+1] + (n+2)f[n], f[n]$]**

Out[2]= $\Big\{ \Big(\frac{3-\sqrt{5}}{2}\Big)^n n^{(1+\sqrt{5})/2}, \Big(\frac{3+\sqrt{5}}{2}\Big)^n n^{(1-\sqrt{5})/2} \Big\}$

For processing this output further, it can at times be more convenient to get it not as an expression. For this case, we offer the possibility to receive the output in an internal format:

In[3]:= **Asymptotics[$(n+1)f[n+2]-(3n+2)f[n+1]+(n+2)f[n], f[n]$, Return → "*internal*"]**

Out[3]= $\Big\{ \Big\{0, \frac{3-\sqrt{5}}{2}, \{\,\}, \frac{1+\sqrt{5}}{2}, \{1\}\Big\}, \Big\{0, \frac{3+\sqrt{5}}{2}, \{\,\}, \frac{1-\sqrt{5}}{2}, \{1\}\Big\} \Big\}$

In this format, a solution

$$\Big(\frac{n}{\mathrm{e}}\Big)^{\gamma n}\exp(\mu_1 n^{1/r}+\mu_2 n^{2/r}+\cdots+\mu_{r-1}n^{1-1/r})\rho^n n^\alpha$$
$$\times\Big(\big(0+\frac{\beta_{0,1}}{n^1}+\cdots+\frac{\beta_{0,N}}{n^N}+\mathrm{O}\big(\frac{1}{n^{N+1}}\big)\big)$$
$$+\cdots+\big(1+\frac{\beta_{k-1,1}}{n^1}+\cdots+\frac{\beta_{k-1,N}}{n^N}+\mathrm{O}\big(\frac{1}{n^{N+1}}\big)\big)\log(n)^{k-1}\Big)$$

is represented in the form

$$\big\{\gamma,\{\mu_1,\ldots,\mu_{r-1}\},\rho,\alpha,\{0,\beta_{0,1},\ldots\},\ldots,\{1,\beta_{k-1,1},\ldots\}\big\}$$

Here is an example with a logarithmic term:

In[4]:= $\mathbf{Asymptotics[(n+2)f[n+2]-(2n+3)f[n+1]+(n+1)f[n],f[n]]}$

Out[4]= $\{\ 1,\ \frac{1}{2n}+\log(n)\ \}$

In[5]:= $\mathbf{Asymptotics[(n{+}2)f[n{+}2]{-}(2n{+}3)f[n{+}1]{+}(n{+}1)f[n],\ f[n],Return\to{}"\mathit{internal}"]}$

Out[5]= $\{\{0,1,\{\},0,\{1\}\},\{0,1,\{\},0,\{0,\frac{1}{2}\},\{1,0\}\}$

By default, the command determines only the dominant terms of the solution. If more terms of the expansion are desired, they can be requested by explicitly specifying the order.

In[6]:= $\mathbf{Asymptotics[(n+1)f[n+2]-(3n+2)f[n+1]+(n+2)f[n],f[n],Order\to 3]}$

Out[6]= $\Big\{\ \Big(\frac{3-\sqrt{5}}{2}\Big)^n n^{(1+\sqrt{5})/2}\Big(1+\frac{10+\sqrt{5}}{10n}+\frac{75-22\sqrt{5}}{120n^2}+\frac{236-107\sqrt{5}}{240n^3}\Big),$
$\Big(\frac{3+\sqrt{5}}{2}\Big)^n n^{(1-\sqrt{5})/2}\Big(1+\frac{10-\sqrt{5}}{10n}+\frac{75+22\sqrt{5}}{120n^2}+\frac{236+107\sqrt{5}}{240n^3}\Big)\ \Big\}$

In[7]:= $\mathbf{Asymptotics[(n+1)f[n+2]-(3n+2)f[n+1]+(n+2)f[n],f[n],Order\to 5]}$

Out[7]= $\Big\{\ \Big(\frac{3-\sqrt{5}}{2}\Big)^n n^{(1+\sqrt{5})/2}\Big(1+\frac{10+\sqrt{5}}{10n}+\frac{75-22\sqrt{5}}{120n^2}+\frac{236-107\sqrt{5}}{240n^3}+\frac{14299-6372\sqrt{5}}{5760n^4}+\frac{2182610-977941\sqrt{5}}{288000n^5}\Big),$
$\Big(\frac{3+\sqrt{5}}{2}\Big)^n n^{(1-\sqrt{5})/2}\Big(1+\frac{10-\sqrt{5}}{10n}+\frac{75+22\sqrt{5}}{120n^2}+\frac{236+107\sqrt{5}}{240n^3}+\frac{14299+6372\sqrt{5}}{5760n^4}+\frac{2182610+977941\sqrt{5}}{288000n^5}\Big)\ \Big\}$

Often, only the higher order terms of one of the solutions are of interest, or one computes with big effort the first terms of an expansion and realizes only afterwards that some additional terms are needed. For these cases, instead of wasting computation time into the computation of uninteresting data, or into the recomputation of data which is already known, there is the second command of the package, **FurtherTerms,** which takes as input one truncated series solution in the internal format and completes it to a truncated series solution of the specified order.

In[8]:= $\mathbf{sols = Asymptotics[(n+1)f[n+2]-(3n+2)f[n+1]+(n+2)f[n],f[n],}$
$\mathbf{Order\to 3, Return\to{}"\mathit{internal}"]}$

Out[8]= $\Big\{\ \Big\{0,\frac{3-\sqrt{5}}{2},\{\,\},\frac{1+\sqrt{5}}{2},\{1,\frac{10+\sqrt{5}}{10},\frac{75-22\sqrt{5}}{120},\frac{236-107\sqrt{5}}{240}\}\Big\},$
$\Big\{0,\frac{3+\sqrt{5}}{2},\{\,\},\frac{1-\sqrt{5}}{2},\{1,\frac{10-\sqrt{5}}{10},\frac{75+22\sqrt{5}}{120},\frac{236+107\sqrt{5}}{240}\}\Big\}\Big\}$

In[9]:= $\mathbf{FurtherTerms[(n+1)f[n+2]-(3n+2)f[n+1]+(n+2)f[n],f[n],sols[[2]],5]}$

Out[9]= $\Big\{0,\frac{3+\sqrt{5}}{2},\{\,\},\frac{1-\sqrt{5}}{2},\{1,\frac{10-\sqrt{5}}{10},\frac{75+22\sqrt{5}}{120},\frac{236+107\sqrt{5}}{240},\frac{14299+6372\sqrt{5}}{5760},\frac{2182610+977941\sqrt{5}}{288000}\}\Big\}$

In[10]:= $\mathbf{FurtherTerms}[(n+1)f[n+2]-(3n+2)f[n+1]+(n+2)f[n],f[n],\%,15]$

Out[10]= $\Big\{0, \frac{3+\sqrt{5}}{2}, \{\,\}, \frac{1-\sqrt{5}}{2}, \{1, \frac{10-\sqrt{5}}{10}, \frac{75+22\sqrt{5}}{120}, \frac{236+107\sqrt{5}}{240}, \frac{14299+6372\sqrt{5}}{5760}, \frac{2182610+977941\sqrt{5}}{288000},$
$\frac{1984202199+886596290\sqrt{5}}{72576000}, \frac{16416301672+7344565027\sqrt{5}}{145152000}, \frac{18418650257095+8235259190664\sqrt{5}}{34836480000},$
$\frac{191695190194310+85740010937053\sqrt{5}}{69672960000}, \frac{65957125132720881+29494472549469338\sqrt{5}}{4180377600000},$
$\frac{412966520815411 8620+1846942946631003389\sqrt{5}}{41803776000000}, \frac{183675686643321718887535+82139222334350902866012\sqrt{5}}{273898340352000000},$
$\frac{2686550605050123303910226+1201493204379434008345249\sqrt{5}}{547796680704000000},$
$\frac{252700769952367203719466267+113009100980753634847297058\sqrt{5}}{6573560168448000000},$
$\frac{105657485223696176457683146240+47252117076548673341670250231\sqrt{5}}{328678008422400000000}\}\Big\}$

3. The Multiplicative Constant

The sequence solutions of a P-finite recurrence equation

$$p_0(n)a_n + p_1(n)a_{n+1} + \cdots + p_r(n)a_{n+r} = 0$$

form a vector space of finite dimension. A particular solution (a_n) can therefore be characterized uniquely by a finite number of initial values $a_0, a_1, \ldots, a_N$. Given such a particular solution, we may wonder about its asymptotic behavior.

Clearly, if we know r linearly independent asymptotic solutions $(b_n^{(1)}), \ldots, (b_n^{(r)})$ (e.g., from the **Asymptotics** command), then we have

$$a_n \sim c_1 b_n^{(1)} + \cdots + c_r b_n^{(r)} \quad (n \to \infty)$$

for certain constants $c_1, \ldots, c_r$. In general, these constants cannot be computed in "closed form", but it is possible to obtain very accurate numerical approximations to them. Here is how.

3.1. **A Single Dominant Term.** First assume for simplicity that the asymptotic solutions $(b_n^{(1)}), \ldots, (b_n^{(r)})$ are such that one of them asymptotically dominates all the others, say $\lim_{n\to\infty} \frac{b_n^{(1)}}{n^k b_n^{(i)}} = 0$ for all $k \in \mathbb{N}$ and $i = 2, \ldots, r$. Then the terms $(b_n^{(2)}), \ldots, (b_n^{(r)})$ are too small to contribute to the asymptotics and we have in fact

$$a_n \sim c_1 b_n^{(1)} \quad (n \to \infty),$$

i.e., there is only a single constant c_1 to be determined. Also if finer asymptotic estimates with higher order terms are considered, only those coming from $(b_n^{(1)})$ will play a role.

In this case, the computation of c_1 is easy. Since we have $\lim_{n\to\infty} \frac{b_n^{(1)}}{a_n} = c_1$, we can obtain decent approximations for c_1 by computing the quotient $b_n^{(1)}/a_n$ for some large index n. The higher the index n, and the more terms of the asymptotic expansion $(b_n^{(1)})$ are taken into account, the more digits of the quotient $b_n^{(1)}/a_n$ will agree with the digits of the actual constant c_1.

In the following typical example session, we define the sequence (a_n) by two initial values and a recurrence of second order. This allows to compute a_n for every specific index n (e.g., for $n = 50$). Next, we determine the dominant terms of the two asymptotic solutions of the recurrence. The second dominates the first, because the basis of its exponential term is greater than in the first. Next, we determine 15 terms of the asymptotic expansion. The approximate value of the constant is finally obtained by computing the quotient $b[n]/a[n]$ for large indices n.

In[11]:= $\boldsymbol{a[0] = 1; a[1] = 2; a[n\_\text{Integer}] := a[n] = ((3(n-2)+2)a[n-1] - na[n-2])/(n-1);}$

In[12]:= $\boldsymbol{a[50]}$

Out[12]= $\dfrac{118654423984991092132736166498746924540704091023070756099945856283 9}{20295399316110442986824002573787115406443832934 4}$

In[13]:= $\textbf{Asymptotics}[(n+1)f[n+2] - (3n+2)f[n+1] + (n+2)f[n], f[n]]$

Out[13]= $\left\{ \left(\frac{3-\sqrt{5}}{2}\right)^n n^{(1+\sqrt{5})/2}, \left(\frac{3+\sqrt{5}}{2}\right)^n n^{(1-\sqrt{5})/2} \right\}$

In[14]:= $\{\mathrm{N}[(3-\sqrt{5})/2], \mathrm{N}[(3+\sqrt{5})/2]\}$

Out[14]= $\{0.381966, 2.61803\}$

In[15]:= $\boldsymbol{terms} = \textbf{Last}[\textbf{FurtherTerms}[(n+1)f[n+2] - (3n+2)f[n+1] + (n+2)f[n], f[n], \{0, \frac{3+\sqrt{5}}{2}, \{\,\}, \frac{1-\sqrt{5}}{2}, \{1\}\}, 15]];$

In[16]:= $b[n\_\text{Integer}] := \left(\frac{3+\sqrt{5}}{2}\right)^n n^{(1-\sqrt{5})/2}\textbf{Sum}[terms[[k+1]]n^{-k}, \{k, 0, 15\}];$

In[17]:= $\textbf{\$RecursionLimit} = 10^5;$

In[18]:= $\mathrm{N}[b[n]/a[n] \ /. \ n \to 500, 50]$

Out[18]= $12.266833273781396098205342055952765464292852811984$

In[19]:= $\mathrm{N}[b[n]/a[n] \ /. \ n \to 5000, 50]$

Out[19]= $12.266833273781396098205342055952770138038305952280$

Which of these digits can we trust? It is instructive to compute the quotient for several indices and see which digits remain fixed. The calculation above strongly suggests that at least the first 30 digits are OK. The convergence is very quick because 15 terms of the expansion were taken into account. This means the error decays to zero in speed $\mathrm{O}(n^{-16})$. More terms will lead to even faster convergence.

The figure below illustrates how for the present example the number of correct decimal digits (vertical axis) grows with n (horizontal axis) when 15 terms of the expansion are used (left) and when 30 terms of the expansion are used (right).





As can be seen from this example, doubling the number of terms in the expansion tends to double the the accuracy of the estimate, while doubling the evaluation index n will usually not double the accuracy.

3.2. **Several Dominant Terms.** When no single term dominates all others, then several constants have to be determined. Two terms are of the same growth for example when the exponential parts have the same absolute value (like 2^n and $(-2)^n$) or when they differ by a polynomial multiple (like 2^n and $n^3 2^n$; although the latter grows more quickly than the former, both terms have to be taken into account because the higher order terms of their expansions may interfere with each other). Suppose that

$$a_n \sim c_1 b_n^{(1)} + \cdots + c_r b_n^{(r)} \quad (n \to \infty)$$

is such that $(b_n^{(1)}), \ldots, (b_n^{(m)})$ are such that $\frac{b_n^{(i)}}{b_n^{(j)}} = \mathrm{O}(n^k)$ for all $1 \leq i, j \leq m$ and some k, and such that $\lim_{n\to\infty} \frac{b_n^{(1)}}{n^k b_n^{(i)}} = 0$ for $i = m+1, \ldots, r$ and all k. Then the first m terms contribute significantly to the asymptotic behavior of (a_n), and the contribution of the remaining terms can be neglected:

$$a_n \sim c_1 b_n^{(1)} + \cdots + c_m b_n^{(m)} \quad (n \to \infty)$$

We have to determine approximations for the constants $c_1, \ldots, c_m$. This can be done by solving a suitable system of linear equations, as illustrated in the following typical example session. In this example, there are two dominant terms: the second and the third. For both of them, we compute the first 15 terms of the expansion. Then we set up a linear system for the desired constants c_2, c_3 and solve it.

In[20]:= $\mathbf{ClearAll}[a, b];$
In[21]:= $a[0] = 2; a[1] = -1; a[2] = 7;$
In[22]:= $a[n\_\mathrm{Integer}] := a[n] = ((40 - 24n + 4n^2)a[-3+n] + (-177 + 120n - 20n^2)a[-2 + n] + (288 - 198n + 33n^2)a[-1+n])/(161 - 108n + 18n^2);$
In[23]:= $a[10]$
Out[23]= $\frac{215985179409378417}{14273864605491407}$
In[24]:= $rec = -4(1+n^2)f[n] + (-3+20n^2)f[1+n] - 3(-3+11n^2)f[2+n] + (-1+18n^2)f[3+n];$
In[25]:= $\mathrm{Asymptotics}[rec, f[n]]$
Out[25]= $\left\{ \left(\frac{1}{2}\right)^n, \left(\frac{2}{3}\right)^n n^{(3-\sqrt{71})/6}, \left(\frac{2}{3}\right)^n n^{(3+\sqrt{71})/6} \right\}$
In[26]:= $terms2 = \mathrm{Last}[\mathrm{FurtherTerms}[rec, f[n], \{0, \frac{2}{3}, \{\}, \frac{3-\sqrt{71}}{6}, \{1\}, 15]];$
In[27]:= $b2[n\_\mathrm{Integer}] := (\frac{2}{3})^n n^{(3-\sqrt{71})/6} \mathrm{Sum}[terms2[[k+1]] n^{-k}, \{k, 0, 15\}];$
In[28]:= $terms3 = \mathrm{Last}[\mathrm{FurtherTerms}[rec, f[n], \{0, \frac{2}{3}, \{\}, \frac{3+\sqrt{71}}{6}, \{1\}, 15]];$
In[29]:= $b3[n\_\mathrm{Integer}] := (\frac{2}{3})^n n^{(3+\sqrt{71})/6} \mathrm{Sum}[terms3[[k+1]] n^{-k}, \{k, 0, 15\}];$
In[30]:= $\$\mathrm{RecursionLimit} = 10^5;$
In[31]:= $A = \mathrm{N}[\{b2[n]/a[n], b3[n]/a[n]\}\ /.\ \{\{n \to 1000\}, \{n \to 2000\}\}, 50];$
In[32]:= $\mathrm{LinearSolve}[A, \{1, 1\}]$
Out[32]= {391.15187878877879622004978270550757884703178680750,
5.0351128647438169746248587210892143931134296563537}
In[33]:= $A = \mathrm{N}[\{b2[n]/a[n], b3[n]/a[n]\}\ /.\ \{\{n \to 2000\}, \{n \to 3000\}\}, 50];$
In[34]:= $\mathrm{LinearSolve}[A, \{1, 1\}]$
Out[34]= {391.15187878877879621999941384812407488923243890678,
5.0351128647438169746248587211162138325823069815638}

The number of correct digits is now less than before, because some accuracy is lost during the linear system solving. But still, the number of correct digits can be increased by taking into account more terms of the expansion. The figure blow shows for this example the number of correct digits in dependence of the evaluation index $(n, n+1000)$ when 15 (left) or 30 (right) terms of the expansions are used.





3.3. **When you need more terms than you can compute.** The best way of getting an accurate estimate for the multiplicative constant(s) is to use as many terms of the expansion as possible. Sometimes, one would like to use even more terms than can be computed explicitly with reasonable effort. For this situation, there exists a simple way to use higher order terms without even knowing them explicitly. This is known as Richardson's convergence acceleration [4]. The idea is that if

$$a_n \sim c\Big(1 + \frac{\beta_k}{n^k} + \frac{\beta_{k+1}}{n^{k+1}} + \cdots\Big) \qquad (n \to \infty)$$

then

$$a_{2n} \sim c\Big(1 + \frac{\beta_k}{2^k n^k} + \frac{\beta_{k+1}}{2^{k+1} n^{k+1}} + \cdots\Big) \qquad (n \to \infty),$$

and therefore

$$\frac{2^k a_{2n} - a_n}{2^k - 1} \sim c\Big(1 + \frac{0}{n^k} + \mathrm{O}(n^{-(k+1)})\Big) \qquad (n \to \infty).$$

This means that $(\frac{2^k a_{2n} - a_n}{2^k - 1})$ converges to the same limit as (a_n), but one order of magnitude faster. Of course, the scheme can be iterated such as to eliminate several terms at once. Here is some Mathematica code for doing this.

```
In[35]:= Richardson[expr_, n_, k_Integer] := Together[(2^k (expr /. n → 2n) − expr)/(2^k − 1)];
In[36]:= Richardson[expr_, n_, {k0_Integer, k1_Integer}] :=
             Fold[Richardson[#1, n, #2]&, expr, Range[k0, k1]]
```

In the following example, we redo the calculation of Section 3.1 by computing only 12 terms of the expansion explicitly and eliminating three more terms with the command just defined.

```
In[37]:= ClearAll[a, b];
In[38]:= a[0] = 1; a[1] = 2; a[n_Integer] := a[n] = ((3(n − 2) + 2)a[n − 1] − na[n − 2])/(n − 1);
In[39]:= terms = Last[FurtherTerms[(n + 1)f[n + 2] − (3n + 2)f[n + 1] + (n + 2)f[n], f[n],
             {0, (3+√5)/2, { }, (1−√5)/2, {1}}, 12]];
In[40]:= b[n_Integer] := ((3+√5)/2)^n n^((1−√5)/2) Sum[terms[[k + 1]]n^−k, {k, 0, 12}];
In[41]:= $RecursionLimit = 10^5;
In[42]:= u[n_Integer] = Richardson[b[n]/a[n], n, {13, 15}];
In[43]:= N[u[500], 50]
Out[43]= 12.266833273781396098205342055952770138038305952602
```

4. Possible Issues

- There seems to be a bug related to the construction of logarithmic terms. When a result is returned, it seems correct, but in some instances, the computation aborts with an error when it should not. This bug will be fixed in a future version.

- In some examples, Mathematica has trouble handling algebraic numbers. Typically, these troubles become more likely for higher order terms. In such situations, it can help to compute only low order terms first, then to rephrase all **Root** expressions in terms of **AlgebraicNumber** expressions of a common number field, and then to apply **FurtherTerms** to this. This is not only more stable but also more efficient.
- Please report other problems to mkauers@risc.jku.at

References

[1] Cyril Banderier, Felix Chern, and Hsien-Kuei Hwang. Asymptotics of D-finite sequences. in preparation.

[2] G.D. Birkhoff and W.J. Trjitzinsky. Analytic theory of singular difference equations. *Acta Mathematica*, 60:1–89, 1932.

[3] Philippe Flajolet and Robert Sedgewick. *Analytic Combinatorics*. Cambridge University Press, 2009.

[4] Lewis Fry Richardson. The approximate arithmetic solution by finite difference of physical problems involving differential equations, with an application to the stress in a masnory dam. *Philosophical Transactions of the Royal Society London, Series A*, 210:307–357, 1910.

[5] Bruno Salvy. Examples of automatic asymptotic expansions. *SIGSAM Bulletin*, 25(2), 1991.

[6] Jet Wimp and Doron Zeilberger. Resurrecting the asymptotics of linear recurrences. *Journal of Mathematical Analysis and Applications*, 111:162–176, 1985.

[7] Doron Zeilberger. Asyrec: A Maple package for computing the asymptotics of solutions of linear recurrence equations with polynomial coefficients. *The Personal Journal of Shalsoh B. Ekhad and Doron Zeilberger*, 2008.

Manuel Kauers, Research Institute for Symbolic Computation, J. Kepler University Linz, Austria

E-mail address: `mkauers@risc.uni-linz.ac.at`